stata.com

binreg — Generalized linear models: Extensions to the binomial family

Description	Quick start	Menu	Syntax
Options	Remarks and examples	Stored results	Methods and formulas
References	Also see		

## Description

binreg fits generalized linear models for the binomial family. It estimates odds ratios, risk ratios, health ratios, and risk differences. The available links are

Option	Implied link	Parameter
or	logit	odds ratios = $\exp(\beta)$
rr	log	risk ratios = $\exp(\beta)$
hr	log complement	health ratios = $\exp(\beta)$
rd	identity	risk differences = $\beta$

Estimates of odds, risk, and health ratios are obtained by exponentiating the appropriate coefficients. The or option produces the same results as Stata's logistic command, and or coefficients yields the same results as the logit command. When no link is specified, or is assumed.

# **Quick start**

Report odds ratios from a model of y on x1 and x2 using a logit link binreg y x1 x2, or

Use the log link and report risk ratios binreg y x1 x2, rr

Use the identity link and report risk differences binreg y x1 x2, rd

Same as above, but with data stored as the number of successes, ys, out of n trials binreg ys x1 x2, rd n(n)

### Menu

Statistics > Generalized linear models > GLM for the binomial family

## Title

# Syntax

binreg depvar [indepvars] [if] [in] [weight] [, options]

options	Description
Model	
<u>nocons</u> tant	suppress constant term
or	use logit link and report odds ratios
rr	use log link and report risk ratios
hr	use log-complement link and report health ratios
rd	use identity link and report risk differences
n(#  <i>varname</i> )	use # or varname for number of trials
exposure(varname)	include ln(varname) in model with coefficient constrained to 1
offset( <i>varname</i> )	include varname in model with coefficient constrained to 1
constraints( <i>constraints</i> )	apply specified linear constraints
mu(varname)	use <i>varname</i> as the initial estimate for the mean of <i>depvar</i>
<u>ini</u> t( <i>varname</i> )	synonym for mu(varname)
SE/Robust	
vce(vcetype)	<i>vcetype</i> may be eim, <u>r</u> obust, <u>cl</u> uster <i>clustvar</i> , oim, opg, <u>boot</u> strap, jackknife, hac <i>kernel</i> , jackknife1, or <u>unb</u> iased
t(varname)	variable name corresponding to time
vfactor(#)	multiply variance matrix by scalar #
 disp(#)	quasilikelihood multiplier
<u>sca</u> le(x2 dev #)	set the scale parameter; default is scale(1)
Reporting	
<u>l</u> evel(#)	set confidence level; default is level(95)
<u>coef</u> ficients	report nonexponentiated coefficients
<u>nocnsr</u> eport	do not display constraints
display_options	control columns and column formats, row spacing, line width, display of omitted variables and base and empty cells, and factor-variable labeling
Maximization	
irls	use iterated, reweighted least-squares optimization; the default
ml	use maximum likelihood optimization
maximize_options	control the maximization process; seldom used
fisher(#)	Fisher scoring steps
search	search for good starting values
<u>col</u> linear	keep collinear variables
<u>coefl</u> egend	display legend instead of statistics

indepvars may contain factor variables; see [U] 11.4.3 Factor variables.

depvar and indepvars may contain time-series operators; see [U] 11.4.4 Time-series varlists.

bayes, bootstrap, by, collect, fp, jackknife, mi estimate, rolling, and statsby are allowed; see [U] 11.1.10 Prefix commands. For more details, see [BAYES] bayes: binreg.

vce(bootstrap), vce(jackknife), and vce(jackknife1) are not allowed with the mi estimate prefix; see [MI] mi estimate.

Weights are not allowed with the bootstrap prefix; see [R] bootstrap.

fweights, iweights, and pweights are allowed; see [U] 11.1.6 weight.

collinear and coeflegend do not appear in the dialog box.

See [U] 20 Estimation and postestimation commands for more capabilities of estimation commands.

#### Options

Model

noconstant; see [R] Estimation options.

or requests the logit link and results in odds ratios if coefficients is not specified.

rr requests the log link and results in risk ratios if coefficients is not specified.

hr requests the log-complement link and results in health ratios if coefficients is not specified.

rd requests the identity link and results in risk differences.

n(# | *varname*) specifies either a constant integer to use as the denominator for the binomial family or a variable that holds the denominator for each observation.

- exposure(*varname*), offset(*varname*), constraints(*constraints*); see [R] Estimation options. constraints(*constraints*) is not allowed with irls.
- mu(*varname*) specifies *varname* containing an initial estimate for the mean of *depvar*. This option can be useful if you encounter convergence difficulties. init(*varname*) is a synonym.

SE/Robust

vce(vcetype) specifies the type of standard error reported, which includes types that are robust to some kinds of misspecification (robust), that allow for intragroup correlation (cluster clustvar), that are derived from asymptotic theory (oim, opg), and that use bootstrap or jackknife methods (bootstrap, jackknife); see [R] vce\_option.

vce(eim), the default, uses the expected information matrix (EIM) for the variance estimator.

binreg also allows the following:

vce(hac kernel [#]) specifies that a heteroskedasticity- and autocorrelation-consistent (HAC) variance estimate be used. HAC refers to the general form for combining weighted matrices to form the variance estimate. There are three kernels built into binreg. kernel is a user-written program or one of

<u>nw</u>est | gallant | <u>an</u>derson

If # is not specified, N - 2 is assumed.

vce(jackknife1) specifies that the one-step jackknife estimate of variance be used.

vce(unbiased) specifies that the unbiased sandwich estimate of variance be used.

#### 4 binreg — Generalized linear models: Extensions to the binomial family

t(varname) specifies the variable name corresponding to time; see [TS] tsset. binreg does not always need to know t(), though it does if vce(hac ...) is specified. Then you can either specify the time variable with t(), or you can tsset your data before calling binreg. When the time variable is required, binreg assumes that the observations are spaced equally over time.

- vfactor(#) specifies a scalar by which to multiply the resulting variance matrix. This option allows users to match output with other packages, which may apply degrees of freedom or other small-sample corrections to estimates of variance.
- disp(#) multiplies the variance of *depvar* by # and divides the deviance by #. The resulting distributions are members of the quasilikelihood family. This option is not allowed with option ml.
- scale(x2|dev|#) overrides the default scale parameter. This option is allowed only with Hessian
  (information matrix) variance estimates.

By default, scale(1) is assumed for the discrete distributions (binomial, Poisson, and negative binomial), and scale(x2) is assumed for the continuous distributions (Gaussian, gamma, and inverse Gaussian).

scale(x2) specifies that the scale parameter be set to the Pearson  $\chi^2$  (or generalized  $\chi^2$ ) statistic divided by the residual degrees of freedom, which is recommended by McCullagh and Nelder (1989) as a good general choice for continuous distributions.

scale(dev) sets the scale parameter to the deviance divided by the residual degrees of freedom. This option provides an alternative to scale(x2) for continuous distributions and overdispersed or underdispersed discrete distributions. This option is not allowed with option ml.

scale(#) sets the scale parameter to #.

Reporting

level(#), noconstant; see [R] Estimation options.

- coefficients displays the nonexponentiated coefficients and corresponding standard errors and confidence intervals. This option has no effect when the rd option is specified, because it always presents the nonexponentiated coefficients.
- nocnsreport; see [R] Estimation options.
- display\_options: noci, nopvalues, noomitted, vsquish, noemptycells, baselevels, allbaselevels, nofvlabel, fvwrap(#), fvwrapon(style), cformat(% fmt), pformat(% fmt), sformat(% fmt), and nolstretch; see [R] Estimation options.

Maximization

irls requests iterated, reweighted least-squares (IRLS) optimization of the deviance instead of Newton-Raphson optimization of the log likelihood. This option is the default.

ml requests that optimization be carried out by using Stata's ml command; see [R] ml.

maximize\_options: difficult, technique(algorithm\_spec), iterate(#), [no]log, trace, gradient, showstep, hessian, showtolerance, tolerance(#), ltolerance(#), nrtolerance(#), nonrtolerance, and from(init\_specs); see [R] Maximize. These options are seldom used.

Setting the optimization method to ml, with technique() set to something other than BHHH, changes the *vcetype* to vce(oim). Specifying technique(bhhh) changes *vcetype* to vce(opg).

Unless option ml is specified, only *maximize\_options* iterate(), nolog, trace, and ltolerance() are allowed. With IRLS optimization, the convergence criterion is satisfied when the absolute change in deviance from one iteration to the next is less than or equal to ltolerance(), where ltolerance(1e-6) is the default.

- fisher (#) specifies the number of Newton-Raphson steps that should use the Fisher scoring Hessian or EIM before switching to the observed information matrix (OIM). This option is available only if ml is specified and is useful only for Newton-Raphson optimization.
- search specifies that the command search for good starting values. This option is available only if ml is specified and is useful only for Newton-Raphson optimization.

The following options are available with binreg but are not shown in the dialog box:

collinear, coeflegend; see [R] Estimation options. collinear is not allowed with irls.

### **Remarks and examples**

Wacholder (1986) suggests methods for estimating risk ratios and risk differences from prospective binomial data. These estimates are obtained by selecting the proper link functions in the generalized linear-model framework. (See *Methods and formulas* for details; also see [R] **glm**.)

#### Example 1

Wacholder (1986) presents an example, using data from Wright et al. (1983), of an investigation of the relationship between alcohol consumption and the risk of a low-birthweight baby. Covariates examined included whether the mother smoked (yes or no), mother's social class (three levels), and drinking frequency (light, moderate, or heavy). The data for the 18 possible categories determined by the covariates are illustrated below.

Let's first describe the data and list a few observations.

```
. use https://www.stata-press.com/data/r18/binreg
(Alcohol and low-birthweight baby)
. list
```

```
category
                    n_lbw_~s
                                 n_women
                                              alcohol
                                                              smokes
                                                                        social
 1.
               1
                           11
                                       84
                                                Heavy
                                                          Nonsmoker
                                                                              1
               2
                            5
 2.
                                       79
                                             Moderate
                                                          Nonsmoker
                                                                              1
 з.
               3
                           11
                                      169
                                                Light
                                                          Nonsmoker
                                                                              1
 4.
               4
                            6
                                       28
                                                Heavy
                                                             Smoker
                                                                              1
 5.
               5
                            3
                                       13
                                             Moderate
                                                             Smoker
                                                                              1
 6.
                                       26
               6
                            1
                                                 Light
                                                             Smoker
                                                                              1
 7.
               7
                            4
                                       22
                                                Heavy
                                                          Nonsmoker
                                                                              2
 8.
               8
                            3
                                                                              2
                                       25
                                                          Nonsmoker
                                             Moderate
               9
                                                                              2
 9.
                           12
                                      162
                                                 Light
                                                          Nonsmoker
                                                                              2
              10
                            4
                                       17
10.
                                                Heavy
                                                             Smoker
11.
              11
                            2
                                        7
                                             Moderate
                                                                              2
                                                             Smoker
12.
              12
                            6
                                       38
                                                             Smoker
                                                                              2
                                                Light
13.
              13
                            0
                                       14
                                                Heavy
                                                          Nonsmoker
                                                                              З
                                                                              з
14.
              14
                            1
                                       18
                                                          Nonsmoker
                                             Moderate
                           12
              15
                                       91
                                                          Nonsmoker
                                                                              З
15.
                                                Light
16.
              16
                            7
                                       19
                                                             Smoker
                                                                              3
                                                Heavy
17.
              17
                            2
                                       18
                                             Moderate
                                                             Smoker
                                                                              3
18.
              18
                            8
                                       70
                                                             Smoker
                                                                              3
                                                Light
```

#### stata.com

Each observation corresponds to one of the 18 covariate structures. The number of low-birthweight babies from n\_women in each category is given by the n\_lbw\_babies variable.

We begin by estimating risk ratios:

. binreg n_lbw_babies i.soc i.alc i.smo, n(n_women) rr						
Iteration 1: Iteration 2: Iteration 3: Iteration 4:	Deviance = :	14.2879 13.607 13.60503 13.60503				
Generalized li Optimization	: MQL Fish (IRLS E			Resid Scale	r of obs = ual df = parameter =	18 12 1
Deviance Pearson	= 13.60 = 11.515				) Deviance = ) Pearson =	
Variance funct Link function		ı*(1-u/n_wom ln(u/n_women		[Bino [Log]	nial]	
				BIC	=	-21.07943
		EIM				
n_lbw_babies	Risk ratio	EIM std. err.	z	P> z	[95% conf.	interval]
n_lbw_babies 	Risk ratio		Z	P> z	[95% conf.	interval]
	Risk ratio		z 1.25	P> z  0.210	[95% conf. .848098	interval] 2.11721
social		std. err.				
social 2	1.340001	std. err.	1.25	0.210	.848098	2.11721
social 2 3	1.340001	std. err.	1.25	0.210	.848098	2.11721
social 2 3 alcohol	1.340001 1.349487	std. err. .3127382 .3291488	1.25	0.210 0.219	.848098 .8366715	2.11721 2.176619
social 2 3 alcohol Moderate	1.340001 1.349487 1.191157	std. err. .3127382 .3291488 .3265354	1.25 1.23 0.64	0.210 0.219 0.523 0.002	.848098 .8366715 .6960276 1.293011	2.11721 2.176619 2.038503
social 2 3 alcohol Moderate Heavy	1.340001 1.349487 1.191157	std. err. .3127382 .3291488 .3265354 .4261751	1.25 1.23 0.64	0.210 0.219 0.523	.848098 .8366715 .6960276	2.11721 2.176619 2.038503

Note: **\_cons** estimates baseline risk.

By default, Stata reports the risk ratios (the exponentiated regression coefficients) estimated by the model. We can see that the risk ratio comparing heavy drinkers with light drinkers, after adjusting for smoking and social class, is 1.974078. That is, mothers who drink heavily during their pregnancy have approximately twice the risk of delivering low-birthweight babies as mothers who are light drinkers.

The nonexponentiated coefficients can be obtained with the coefficients option:

. binreg n_lbw	_babies i.soc i.alc i.smo, n(	(n_women) rr coefficients	
Iteration 2: Iteration 3:	Deviance = 14.2879 Deviance = 13.607 Deviance = 13.60503 Deviance = 13.60503		
Generalized li	inear models	Number of obs =	18
Optimization	: MQL Fisher scoring	Residual df =	12
Deviance Pearson	(IRLS EIM) = 13.6050268 = 11.51517095	Scale parameter = (1/df) Deviance = (1/df) Pearson =	1.133752
	cion: $V(u) = u*(1-u/n_women)$ : $g(u) = ln(u/n_women)$	[Binomial] [Log]	
		BIC =	-21.07943
	EIM		

		EIM				
n_lbw_babies	Coefficient	std. err.	z	P> z	[95% conf.	interval]
social						
2	.2926702	.2333866	1.25	0.210	1647591	.7500994
3	. 2997244	.2439066	1.23	0.219	1783238	.7777726
alcohol						
Moderate	.1749248	.274133	0.64	0.523	362366	.7122156
Heavy	.6801017	.2158856	3.15	0.002	.2569737	1.10323
smokes						
Smoker	.4998317	.2019329	2.48	0.013	.1040505	.8956129
_cons	-2.764079	.2031606	-13.61	0.000	-3.162266	-2.365891

Risk differences are obtained with the rd option:

Iteration 1: Deviance = 18.67277	
Iteration 2: Deviance = 14.94364	
Iteration 3: Deviance = 14.9185	
Iteration 4: Deviance = 14.91762	
Iteration 5: Deviance = 14.91758	
Iteration 6: Deviance = 14.91758	
Iteration 7: Deviance = 14.91758	
Generalized linear models Number of obs =	18
Optimization : MQL Fisher scoring Residual df =	12
(IRLS EIM) Scale parameter =	1
Deviance = 14.91758277 (1/df) Deviance = 1.243	132
Pearson = 12.60353235 (1/df) Pearson = 1.0502	294
Variance function: V(u) = u*(1-u/n_women) [Binomial]	
Link function : g(u) = u/n_women [Identity]	
BIC = -19.76	388
EIM	
n_lbw_babies Risk diff. std. err. z P> z  [95% conf. interva	1]
social	
2 .0263817 .0232124 1.14 0.2560191137 .0718	71
3 .0365553 .0268668 1.36 0.1740161026 .0892	132
alcohol	
Moderate .0122539 .0257713 0.48 0.6340382569 .06276	
Heavy .0801291 .0302878 2.65 0.008 .020766 .1394	921
smokes	
Smoker .0542415 .0270838 2.00 0.045 .0011582 .10732	248
_cons .059028 .0160693 3.67 0.000 .0275327 .09052	232

The risk difference between heavy drinkers and light drinkers is 0.0801291. Because the risk differences are obtained directly from the coefficients estimated by using the identity link, the coefficients option would have no effect here.

Health ratios are obtained with the hr option. The health ratios (exponentiated coefficients for the log-complement link) are reported directly.

. binreg n\_lbw\_babies i.soc i.alc i.smo, n(n\_women) hr Iteration 1: Deviance = 21.15233 Iteration 2: Deviance = 15.16467 Iteration 3: Deviance = 15.13205 Iteration 4: Deviance = 15.13114 Iteration 5: Deviance = 15.13111 Iteration 6: Deviance = 15.13111 Iteration 7: Deviance = 15.13111 Generalized linear models Number of obs = 18 Optimization : MQL Fisher scoring Residual df = 12 (IRLS EIM) Scale parameter = 1 Deviance = 15.13110545 (1/df) Deviance = 1.260925 Pearson = 12.84203917 (1/df) Pearson = 1.07017 Variance function: V(u) = u\*(1-u/n women)[Binomial] Link function :  $g(u) = ln(1-u/n_women)$ [Log complement] BTC = -19.55336 EIM n\_lbw\_babies Hlth ratio std. err. z P>|z| [95% conf. interval] social

social						
2	.9720541	.024858	-1.11	0.268	.9245342	1.022017
3	.9597182	.0290412	-1.36	0.174	.9044535	1.01836
alcohol						
Moderate	.9871517	.0278852	-0.46	0.647	.9339831	1.043347
Heavy	.9134243	.0325726	-2.54	0.011	.8517631	.9795493
smokes						
Smoker	.9409983	.0296125	-1.93	0.053	.8847125	1.000865
_cons	.9409945	.0163084	-3.51	0.000	.9095674	.9735075

Note: \_cons estimates baseline health (probability of no disease).

To see the nonexponentiated coefficients, we could specify the coefficients option.

# **Stored results**

binreg, irls stores the following in e():

Scalars	
e(N)	number of observations
e(k)	number of parameters
e(k_eq_model)	number of equations in overall model test
e(df_m)	model degrees of freedom
e(df)	residual degrees of freedom
e(phi)	model scale parameter
e(disp)	dispersion parameter
e(bic)	model BIC
e(N_clust)	number of clusters
e(deviance)	deviance
e(deviance_s)	scaled deviance
e(deviance_p)	Pearson deviance
e(deviance_ps)	scaled Pearson deviance
e(dispers)	dispersion
e(dispers_s)	scaled dispersion
e(dispers_p)	Pearson dispersion
	scaled Pearson dispersion
e(dispers_ps) e(vf)	
	factor set by vfactor(), 1 if not set
e(rank)	rank of e(V)
e(rc)	return code
Macros	
e(cmd)	binreg
e(cmdline)	command as typed
e(depvar)	name of dependent variable
e(eform)	eform() option implied by or, rr, hr, or rd
e(varfunc)	program to calculate variance function
e(varfunct)	variance title
e(varfuncf)	variance function
e(link)	program to calculate link function
e(linkt)	link title
e(linkf)	link function
e(m)	number of binomial trials
e(wtype)	weight type
e(wexp)	weight expression
e(title)	title in estimation output
e(title_fl)	family–link title
e(clustvar)	name of cluster variable
e(offset)	linear offset variable
e(cons)	noconstant or not set
e(hac_kernel)	HAC kernel
e(hac_lag)	HAC lag
e(vce)	vcetype specified in vce()
	title used to label Std. err.
e(vcetype)	type of optimization
e(opt)	•••
e(opt1)	optimization title, line 1
e(opt2)	optimization title, line 2
e(properties)	b V
e(predict)	program used to implement predict
e(marginsok)	predictions allowed by margins
e(marginsnotok)	predictions disallowed by margins
e(asbalanced)	factor variables fvset as asbalanced
e(asobserved)	factor variables fvset as asobserved

Matrices e(b) e(V) e(V_modelbased)	coefficient vector variance–covariance matrix of the estimators model-based variance
Functions e(sample)	marks estimation sample

In addition to the above, the following is stored in r():

Matrices r(table)

matrix containing the coefficients with their standard errors, test statistics, *p*-values, and confidence intervals

Note that results stored in r() are updated when the command is replayed and will be replaced when any r-class command is run after the estimation command.

binreg, ml stores the following in e():

•		e
Scalars		
e(N)		number of observations
e(k)		number of parameters
e(k_	eq)	number of equations in e(b)
e(k_	eq_model)	number of equations in overall model test
e(k_0	dv)	number of dependent variables
e(df.	_m)	model degrees of freedom
e(df)	)	residual degrees of freedom
e(ph:	i)	model scale parameter
e(ai	c)	model AIC, if ml
e(bio	c)	model BIC
e(11)	)	log likelihood, if ml
e(N_	clust)	number of clusters
e(ch:	i2)	$\chi^2$
e(p)		<i>p</i> -value for model test
e(de	/iance)	deviance
e(de	/iance_s)	scaled deviance
e(de	/iance_p)	Pearson deviance
e(de	/iance_ps)	scaled Pearson deviance
e(di:	spers)	dispersion
e(di:	spers_s)	scaled dispersion
e(di:	spers_p)	Pearson dispersion
e(di:	spers_ps)	scaled Pearson dispersion
e(vf)	)	factor set by vfactor(), 1 if not set
e(ra	ık)	rank of e(V)
e(ic)	)	number of iterations
e(rc)	)	return code
e(co	nverged)	1 if converged, 0 otherwise
Macros		
e(cmo	1)	binreg
e(cmo	dline)	command as typed
e(de	ovar)	name of dependent variable
e(ef	orm)	eform() option implied by or, rr, hr, or rd
e(va	func)	program to calculate variance function
e(va	funct)	variance title
e(va	funcf)	variance function
e(li	ık)	program to calculate link function
e(li	nkt)	link title
e(li	nkf)	link function
e(m)		number of binomial trials
e(wt	/pe)	weight type
e(we	cp)	weight expression
e(ti	tle)	title in estimation output

<pre>e(title_fl) e(clustvar) e(offset) e(cons) e(hac_kernel) e(hac_lag) e(chi2type) e(vce) e(vceype) e(opt) e(opt1) e(which) e(ml_method) e(user)</pre>	family-link title name of cluster variable linear offset variable noconstant or not set HAC kernel HAC lag Wald; type of model $\chi^2$ test <i>vcetype</i> specified in vce() title used to label Std. err. type of optimization optimization title, line 1 max or min; whether optimizer is to perform maximization or minimization type of ml method name of likelihood-evaluator program
e(technique) e(properties) e(predict) e(marginsok) e(marginsnotok) e(asbalanced)	maximization technique b V program used to implement predict predictions allowed by margins predictions disallowed by margins factor variables fvset as asbalanced
e(asobserved)	factor variables fvset as asobserved
Matrices e(b) e(Cns) e(ilog) e(gradient) e(V) e(V_modelbased)	coefficient vector constraints matrix iteration log (up to 20 iterations) gradient vector variance–covariance matrix of the estimators model-based variance
Functions e(sample)	marks estimation sample

In addition to the above, the following is stored in r():

Matrices r(table)

matrix containing the coefficients with their standard errors, test statistics, *p*-values, and confidence intervals

Note that results stored in r() are updated when the command is replayed and will be replaced when any r-class command is run after the estimation command.

#### Methods and formulas

Let  $\pi_i$  be the probability of success for the *i*th observation, i = 1, ..., N, and let  $X\beta$  be the linear predictor. The link function relates the covariates of each observation to its respective probability through the linear predictor.

In logistic regression, the logit link is used:

$$\ln\!\left(\frac{\pi}{1-\pi}\right) = X\beta$$

The regression coefficient  $\beta_k$  represents the change in the logarithm of the odds associated with a one-unit change in the value of the  $X_k$  covariate; thus,  $\exp(\beta_k)$  is the ratio of the odds associated with a change of one unit in  $X_k$ .

For risk differences, the identity link  $\pi = X\beta$  is used. The regression coefficient  $\beta_k$  represents the risk difference associated with a change of one unit in  $X_k$ . When using the identity link, you can obtain fitted probabilities outside the interval (0, 1). As suggested by Wacholder, at each iteration,

fitted probabilities are checked for range conditions (and put back in range if necessary). For example, if the identity link results in a fitted probability that is smaller than 1e–4, the probability is replaced with 1e–4 before the link function is calculated.

A similar adjustment is made for the logarithmic link, which is used for estimating the risk ratio,  $\ln(\pi) = X\beta$ , where  $\exp(\beta_k)$  is the risk ratio associated with a change of one unit in  $X_k$ , and for the log-complement link used to estimate the probability of no disease or health, where  $\exp(\beta_k)$ represents the "health ratio" associated with a change of one unit in  $X_k$ .

This command supports the Huber/White/sandwich estimator of the variance and its clustered version using vce(robust) and vce(cluster *clustvar*), respectively. See [P] **\_robust**, particularly *Maximum likelihood estimators* and *Methods and formulas*.

## References

- Cummings, P. 2009. Methods for estimating adjusted risk ratios. Stata Journal 9: 175-196.
- Kleinbaum, D. G., and M. Klein. 2010. Logistic Regression: A Self-Learning Text. 3rd ed. New York: Springer.
- McCullagh, P., and J. A. Nelder. 1989. Generalized Linear Models. 2nd ed. London: Chapman and Hall/CRC.
- Wacholder, S. 1986. Binomial regression in GLIM: Estimating risk ratios and risk differences. American Journal of Epidemiology 123: 174–184. https://doi.org/10.1093/oxfordjournals.aje.a114212.
- Wright, J. T., I. G. Barrison, I. G. Lewis, K. D. MacRae, E. J. Waterson, P. J. Toplis, M. G. Gordon, N. F. Morris, and I. M. Murray-Lyon. 1983. Alcohol consumption, pregnancy and low birthweight. *Lancet* 1: 663–665. https://doi.org/10.1016/S0140-6736(83)91964-5.

#### Also see

- [R] **binreg postestimation** Postestimation tools for binreg
- [R] **glm** Generalized linear models
- [BAYES] bayes: binreg Bayesian generalized linear models: Extensions to the binomial family
- [ME] mecloglog Multilevel mixed-effects complementary log-log regression
- [ME] meglm Multilevel mixed-effects generalized linear models
- [ME] melogit Multilevel mixed-effects logistic regression
- [ME] meprobit Multilevel mixed-effects probit regression
- [MI] Estimation Estimation commands for use with mi estimate
- [U] 20 Estimation and postestimation commands

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on citing Stata documentation.